EHands: Quantum Protocol for Polynomial Computation on Real-Valued Encoded States

Mathematica notebook for arXiv:2502.15928
October 4, 2025

Quantum Gates

In[ ]:=
```
X = {{0, 1}, {1, 0}};
Y = {{0, -I}, {I, 0}};
Z = {{1, 0}, {0, -1}};
H = {{1, 1}, {1, -1}} / Sqrt[2];
```

In[ ]:=
```
MatrixForm[X];
MatrixForm[Y];
MatrixForm[Z];
MatrixForm[H]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} \end{pmatrix}$$

In[ ]:=
```
Rx[theta_] := {{Cos[theta / 2], -I * Sin[theta / 2]}, {-I * Sin[theta / 2], Cos[theta / 2]}};
Ry[theta_] := {{Cos[theta / 2], -Sin[theta / 2]}, {Sin[theta / 2], Cos[theta / 2]}};
Rz[theta_] := {{Exp[-I * theta / 2], 0}, {0, Exp[I * theta / 2]}};
```

```
In[•]:= MatrixForm[Rx[t]];
        MatrixForm[Ry[t]]
        MatrixForm[Rz[t]];
```

Out[•]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}\left[\frac{t}{2}\right] & -\mathrm{Sin}\left[\frac{t}{2}\right] \\ \mathrm{Sin}\left[\frac{t}{2}\right] & \mathrm{Cos}\left[\frac{t}{2}\right] \end{pmatrix}$$

```
In[•]:= myRy[x_] := Ry[t_] /. Cos[t_ / 2] :> Sqrt[(1 + x) / 2] /. Sin[t_ / 2] :> Sqrt[(1 - x) / 2];
        MatrixForm[myRy[x]]
```

Out[•]//MatrixForm=

$$\begin{pmatrix} \frac{\sqrt{1+x}}{\sqrt{2}} & -\frac{\sqrt{1-x}}{\sqrt{2}} \\ \frac{\sqrt{1-x}}{\sqrt{2}} & \frac{\sqrt{1+x}}{\sqrt{2}} \end{pmatrix}$$

```
In[•]:= CNOT01 = {{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 0, 1}, {0, 0, 1, 0}};
        CNOT10 = {{1, 0, 0, 0}, {0, 0, 0, 1}, {0, 0, 1, 0}, {0, 1, 0, 0}};
        CZ = {{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, -1}};
        SWAP = {{1, 0, 0, 0}, {0, 0, 1, 0}, {0, 1, 0, 0}, {0, 0, 0, 1}};
```

```
In[•]:= MatrixForm[CNOT01];
        MatrixForm[CNOT10];
        MatrixForm[CZ];
        MatrixForm[SWAP];
```

```
In[•]:= CNOT02 = BlockDiagonalMatrix[{IdentityMatrix[4], X, X}];
        tmp1 = {{1, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 0}};
        tmp2 = {{0, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 1}};
        CNOT20 = ArrayFlatten[{{tmp1, tmp2}, {tmp2, tmp1}}];
```

```
In[•]:= MatrixForm[CNOT02];
        MatrixForm[CNOT20];
```

Encoding and decoding of 2 real numbers on 2 qubits

In[ ]:=
```
Uprep = KroneckerProduct[myRy[x0], myRy[x1]];
ket00 = IdentityMatrix[{4, 1}];
phi = Uprep.ket00;
MatrixForm[phi]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2}\sqrt{1+x0}\sqrt{1+x1} \\ \frac{1}{2}\sqrt{1+x0}\sqrt{1-x1} \\ \frac{1}{2}\sqrt{1-x0}\sqrt{1+x1} \\ \frac{1}{2}\sqrt{1-x0}\sqrt{1-x1} \end{pmatrix}$$

In[ ]:=
```
Os = KroneckerProduct[Z, IdentityMatrix[2]];
Op = KroneckerProduct[IdentityMatrix[2], Z];
```

In[ ]:=
```
MatrixForm[Os]
MatrixForm[Op]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

In[ ]:=
```
Simplify[ConjugateTranspose[phi].Os.phi, {-1 ≤ x0 ≤ 1, -1 ≤ x1 ≤ 1}]
Simplify[ConjugateTranspose[phi].Op.phi, {-1 ≤ x0 ≤ 1, -1 ≤ x1 ≤ 1}]
```

Out[ ]=

{{x0}}

Out[ ]=

{{x1}}

Product-with-memory circuit

In[∘]:=
```
MatrixForm[Rz[Pi / 2]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} e^{-\frac{i\pi}{4}} & 0 \\ 0 & e^{\frac{i\pi}{4}} \end{pmatrix}$$

In[∘]:=
```
Uprod = CNOT01.KroneckerProduct[IdentityMatrix[2], Rz[Pi / 2]];
MatrixForm[Uprod / Exp[-I * Pi / 4]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & i & 0 & 0 \\ 0 & 0 & 0 & i \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

In[∘]:=
```
psiP = Uprod.phi;
MatrixForm[psiP / (Exp[-I * Pi / 4] / 2)]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \sqrt{1 + x0}\ \sqrt{1 + x1} \\ i\ \sqrt{1 + x0}\ \sqrt{1 - x1} \\ i\ \sqrt{1 - x0}\ \sqrt{1 - x1} \\ \sqrt{1 - x0}\ \sqrt{1 + x1} \end{pmatrix}$$

In[∘]:=
```
Simplify[ConjugateTranspose[psiP].Os.psiP, {-1 ≤ x0 ≤ 1, -1 ≤ x1 ≤ 1}]
Simplify[ConjugateTranspose[psiP].Op.psiP, {-1 ≤ x0 ≤ 1, -1 ≤ x1 ≤ 1}]
```

Out[∘]=

{{x0}}

Out[∘]=

{{x0 x1}}

Weighted sum circuit

In[∘]:=
```
alpha[w_] := ArcCos[1 - 2 * w];
alpha[w]
```

Out[∘]=
```
ArcCos[1 - 2 w]
```

In[∘]:=
```
Rypos[theta_] := Ry[t] /. Cos[t_ / 2] :→ Sqrt[(1 + Cos[theta]) / 2] /. Sin[t_ / 2] :→ Sqrt[(1 - Cos[theta]) / 2];
Ryneg[theta_] := Ry[t] /. Cos[t_ / 2] :→ Sqrt[(1 + Cos[theta]) / 2] /. Sin[t_ / 2] :→ -Sqrt[(1 - Cos[theta]) / 2];
MatrixForm[Rypos[x]];
MatrixForm[Ryneg[x]];
myRypos2[w_] := Rypos[alpha[t_] / 2] /. Cos[ArcCos[1 - 2 * t_] / 2] :→ Simplify[Sqrt[(1 + Cos[ArcCos[1 - 2 * w]]) / 2]];
myRyneg2[w_] := Ryneg[alpha[t_] / 2] /. Cos[ArcCos[1 - 2 * t_] / 2] :→ Simplify[Sqrt[(1 + Cos[ArcCos[1 - 2 * w]]) / 2]];
MatrixForm[Sqrt[2] * myRypos2[w]]
MatrixForm[Sqrt[2] * myRyneg2[w]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \sqrt{1+\sqrt{1-w}} & -\sqrt{1-\sqrt{1-w}} \\ \sqrt{1-\sqrt{1-w}} & \sqrt{1+\sqrt{1-w}} \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} \sqrt{1+\sqrt{1-w}} & \sqrt{1-\sqrt{1-w}} \\ -\sqrt{1-\sqrt{1-w}} & \sqrt{1+\sqrt{1-w}} \end{pmatrix}$$

In[∘]:=
```
Usum[w_] := Simplify[KroneckerProduct[myRyneg2[w], IdentityMatrix[2]].
    CNOT10.KroneckerProduct[myRypos2[w], IdentityMatrix[2]].Uprod];
MatrixForm[Simplify[Usum[w] / Exp[-I * Pi / 4]]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & i\sqrt{w} & \sqrt{1-w} & 0 \\ 0 & 0 & 0 & i \\ 0 & i\sqrt{1-w} & -\sqrt{w} & 0 \end{pmatrix}$$

In[∘]:=
```
psiS = Usum[w].phi;
MatrixForm[Simplify[psiS / (Exp[-I * Pi / 4] / 2)]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \sqrt{1+x0}\ \sqrt{1+x1} \\ i\sqrt{w}\ \sqrt{1+x0}\ \sqrt{1-x1} + \sqrt{1-w}\ \sqrt{1-x0}\ \sqrt{1+x1} \\ i\sqrt{1-x0}\ \sqrt{1-x1} \\ i\sqrt{1-w}\ \sqrt{1+x0}\ \sqrt{1-x1} - \sqrt{w}\ \sqrt{1-x0}\ \sqrt{1+x1} \end{pmatrix}$$

In[∘]:=
```
Simplify[ConjugateTranspose[psiS].Os.psiS, {-1 ≤ x0 ≤ 1, -1 ≤ x1 ≤ 1, 0 ≤ w ≤ 1}]
Simplify[ConjugateTranspose[psiS].Op.psiS, {-1 ≤ x0 ≤ 1, -1 ≤ x1 ≤ 1, 0 ≤ w ≤ 1}]
```

Out[∘]=

{{w (x0 - x1) + x1}}

Out[∘]=

{{x0 x1}}

Linear combinations by concatenating summation circuits

In[∘]:=
```
Uprep3 = KroneckerProduct[myRy[x0], myRy[x1], myRy[x2]];
ket000 = IdentityMatrix[{8, 1}];
phi3 = Uprep3.ket000;
MatrixForm[phi3 * 2 * Sqrt[2]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \sqrt{1+x0}\ \sqrt{1+x1}\ \sqrt{1+x2} \\ \sqrt{1+x0}\ \sqrt{1+x1}\ \sqrt{1-x2} \\ \sqrt{1+x0}\ \sqrt{1-x1}\ \sqrt{1+x2} \\ \sqrt{1+x0}\ \sqrt{1-x1}\ \sqrt{1-x2} \\ \sqrt{1-x0}\ \sqrt{1+x1}\ \sqrt{1+x2} \\ \sqrt{1-x0}\ \sqrt{1+x1}\ \sqrt{1-x2} \\ \sqrt{1-x0}\ \sqrt{1-x1}\ \sqrt{1+x2} \\ \sqrt{1-x0}\ \sqrt{1-x1}\ \sqrt{1-x2} \end{pmatrix}$$

In[ ]:=
```
Upflip = CZ.KroneckerProduct[H, IdentityMatrix[2]];
MatrixForm[Upflip]
Usum2[w_] := Simplify[KroneckerProduct[myRyneg2[w], IdentityMatrix[4]].CNOT20.
    KroneckerProduct[myRypos2[w], IdentityMatrix[4]].CNOT02.KroneckerProduct[IdentityMatrix[4], Rz[Pi / 2]]];
Usum3[w0_, w1_] :=
  Simplify[KroneckerProduct[IdentityMatrix[2], Usum2[w1]].KroneckerProduct[Upflip, IdentityMatrix[4]].
    KroneckerProduct[IdentityMatrix[2], Usum[w0], IdentityMatrix[2]]];
MatrixForm[Usum3[w0, w1]]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{1}{\sqrt{2}} & 0 & \frac{1}{\sqrt{2}} & 0 \\ 0 & \frac{1}{\sqrt{2}} & 0 & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & 0 & -\frac{1}{\sqrt{2}} & 0 \\ 0 & -\frac{1}{\sqrt{2}} & 0 & \frac{1}{\sqrt{2}} \end{pmatrix}$$

Out[ ]//MatrixForm=

$$\begin{pmatrix}
-\frac{i}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -\frac{i}{\sqrt{2}} & 0 & 0 & 0 & 0 \\
0 & \frac{\sqrt{w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & \frac{\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & \frac{\sqrt{w1}}{\sqrt{2}} & 0 & 0 & 0 \\
0 & 0 & \frac{\sqrt{w0}}{\sqrt{2}} & 0 & -\frac{i\sqrt{1-w0}}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 & \frac{\sqrt{w0}}{\sqrt{2}} & 0 & -\frac{i\sqrt{1-w0}}{\sqrt{2}} \\
0 & 0 & \frac{\sqrt{1-w0}\sqrt{1-w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{1-w1}}{\sqrt{2}} & \frac{\sqrt{1-w0}\sqrt{w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & \frac{\sqrt{1-w0}\sqrt{1-w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{1-w}}{\sqrt{2}} \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{i}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 \\
0 & \frac{\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & -\frac{\sqrt{w1}}{\sqrt{2}} & 0 & 0 & \frac{\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & 0 \\
0 & 0 & 0 & \frac{i\sqrt{1-w0}}{\sqrt{2}} & 0 & -\frac{\sqrt{w0}}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 & \frac{i\sqrt{1-w0}}{\sqrt{2}} & 0 \\
0 & 0 & -\frac{\sqrt{1-w0}\sqrt{w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{1-w1}}{\sqrt{2}} & -\frac{i\sqrt{w0}\sqrt{w1}}{\sqrt{2}} & \frac{\sqrt{1-w0}\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & -\frac{\sqrt{1-w0}\sqrt{w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{1-w1}}{\sqrt{2}} & -\frac{i\sqrt{w0}\sqrt{w}}{\sqrt{2}} \\
-\frac{i}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{i}{\sqrt{2}} & 0 & 0 & 0 & 0 \\
0 & \frac{\sqrt{w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & -\frac{\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & -\frac{\sqrt{w1}}{\sqrt{2}} & 0 & 0 & 0 \\
0 & 0 & \frac{\sqrt{w0}}{\sqrt{2}} & 0 & -\frac{i\sqrt{1-w0}}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 & -\frac{\sqrt{w0}}{\sqrt{2}} & 0 & \frac{i\sqrt{1-w0}}{\sqrt{2}} \\
0 & 0 & -\frac{\sqrt{1-w0}\sqrt{1-w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{w1}}{\sqrt{2}} & -\frac{i\sqrt{w0}\sqrt{1-w1}}{\sqrt{2}} & \frac{\sqrt{1-w0}\sqrt{w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & \frac{\sqrt{1-w0}\sqrt{1-w1}}{\sqrt{2}} & -\frac{i\sqrt{w0}\sqrt{w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{1-w}}{\sqrt{2}} \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & -\frac{i}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 \\
0 & \frac{\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & \frac{\sqrt{w1}}{\sqrt{2}} & 0 & 0 & -\frac{\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & 0 \\
0 & 0 & 0 & -\frac{i\sqrt{1-w0}}{\sqrt{2}} & 0 & \frac{\sqrt{w0}}{\sqrt{2}} & 0 & 0 & 0 & 0 & 0 & \frac{i\sqrt{1-w0}}{\sqrt{2}} & 0 \\
0 & 0 & \frac{\sqrt{1-w0}\sqrt{w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{1-w1}}{\sqrt{2}} & \frac{i\sqrt{w0}\sqrt{w1}}{\sqrt{2}} & \frac{\sqrt{1-w0}\sqrt{1-w1}}{\sqrt{2}} & 0 & 0 & 0 & 0 & -\frac{\sqrt{1-w0}\sqrt{w1}}{\sqrt{2}} & -\frac{i\sqrt{w0}\sqrt{1-w1}}{\sqrt{2}} & -\frac{i\sqrt{w0}\sqrt{w}}{\sqrt{2}}
\end{pmatrix}$$

In[∘]:=
```
ket0 = IdentityMatrix[{2, 1}];
psi3 = Usum3[w0, w1].KroneckerProduct[ket0, phi3];
MatrixForm[Simplify[psi3]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix}
-\frac{1}{4}\,\mathrm{i}\,\sqrt{1+x0}\,\sqrt{1+x1}\,\sqrt{1+x2} \\
\frac{1}{4}\left(\sqrt{w1}\,\sqrt{1+x0}\,\sqrt{1+x1}\,\sqrt{1-x2}+\sqrt{1-w1}\,\sqrt{1-x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right) \\
\frac{1}{4}\left(\sqrt{w0}\,\sqrt{1+x0}\,\sqrt{1-x1}-\mathrm{i}\,\sqrt{1-w0}\,\sqrt{1-x0}\,\sqrt{1+x1}\right)\sqrt{1+x2} \\
\frac{1}{4}\left(\sqrt{1-w0}\left(\sqrt{w1}\,\sqrt{1-x0}\,\sqrt{1+x1}\,\sqrt{1-x2}+\sqrt{1-w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right)+\mathrm{i}\,\sqrt{w0}\left(\sqrt{w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1-x2}+\sqrt{1-w1}\cdot\right.\right. \\
\frac{1}{4}\,\mathrm{i}\,\sqrt{1-x0}\,\sqrt{1-x1}\,\sqrt{1-x2} \\
\frac{1}{4}\left(\sqrt{1-w1}\,\sqrt{1+x0}\,\sqrt{1+x1}\,\sqrt{1-x2}-\sqrt{w1}\,\sqrt{1-x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right) \\
\frac{1}{4}\,\mathrm{i}\left(\sqrt{1-w0}\,\sqrt{1+x0}\,\sqrt{1-x1}+\mathrm{i}\,\sqrt{w0}\,\sqrt{1-x0}\,\sqrt{1+x1}\right)\sqrt{1-x2} \\
\frac{1}{4}\left(\sqrt{1-w0}\left(\sqrt{1-w1}\,\sqrt{1-x0}\,\sqrt{1+x1}\,\sqrt{1-x2}-\sqrt{w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right)+\mathrm{i}\,\sqrt{w0}\left(\sqrt{1-w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1-x2}-\sqrt{w1}\cdot\right.\right. \\
-\frac{1}{4}\,\mathrm{i}\,\sqrt{1+x0}\,\sqrt{1+x1}\,\sqrt{1+x2} \\
\frac{1}{4}\left(\sqrt{w1}\,\sqrt{1+x0}\,\sqrt{1+x1}\,\sqrt{1-x2}-\sqrt{1-w1}\,\sqrt{1-x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right) \\
\frac{1}{4}\left(\sqrt{w0}\,\sqrt{1+x0}\,\sqrt{1-x1}-\mathrm{i}\,\sqrt{1-w0}\,\sqrt{1-x0}\,\sqrt{1+x1}\right)\sqrt{1+x2} \\
\frac{1}{4}\left(\sqrt{1-w0}\left(\sqrt{w1}\,\sqrt{1-x0}\,\sqrt{1+x1}\,\sqrt{1-x2}-\sqrt{1-w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right)-\mathrm{i}\,\sqrt{w0}\left(-\sqrt{w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1-x2}+\sqrt{1-w1}\right.\right. \\
-\frac{1}{4}\,\mathrm{i}\,\sqrt{1-x0}\,\sqrt{1-x1}\,\sqrt{1-x2} \\
\frac{1}{4}\left(\sqrt{1-w1}\,\sqrt{1+x0}\,\sqrt{1+x1}\,\sqrt{1-x2}+\sqrt{w1}\,\sqrt{1-x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right) \\
\frac{1}{4}\left(-\mathrm{i}\,\sqrt{1-w0}\,\sqrt{1+x0}\,\sqrt{1-x1}+\sqrt{w0}\,\sqrt{1-x0}\,\sqrt{1+x1}\right)\sqrt{1-x2} \\
\frac{1}{4}\left(\sqrt{1-w0}\left(\sqrt{1-w1}\,\sqrt{1-x0}\,\sqrt{1+x1}\,\sqrt{1-x2}+\sqrt{w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1+x2}\right)+\mathrm{i}\,\sqrt{w0}\left(\sqrt{1-w1}\,\sqrt{1+x0}\,\sqrt{1-x1}\,\sqrt{1-x2}+\sqrt{w1}\cdot\right.\right.
\end{pmatrix}$$

In[∘]:=
```
Os3 = KroneckerProduct[IdentityMatrix[2], Z, IdentityMatrix[4]];
FullSimplify[ConjugateTranspose[psi3].Os3.psi3, {-1 ≤ x0 ≤ 1, -1 ≤ x1 ≤ 1, -1 ≤ x2 ≤ 1, 0 ≤ w0 ≤ 1, 0 ≤ w1 ≤ 1}]
```

Out[∘]=

{{w1 (w0 (x0 - x1) + x1 - x2) + x2}}